Day 1: Introduction

We talked today about the massive transformation that electricity markets are witnessing, with the rapid growth of renewable power and explicity goal of fully decarbonizing the electricity market in coming years.

In this practice session, we will examine time series data from the Spanish electricity market, which has substantial intermittent renewable power (wind and solar).

The data have been collected from publicly available sources (Red Eléctrica de España and OMIE, among others). The data are from the paper "Measuring the Impact of Wind Power in the Spanish Electricity Market," by Claire Petersen, Mar Reguant, and Lola Segura.

We need to load packages in Julia, similar to the import function in Python or the library functionality in R. *Pluto* will install the libraries automatically, but in Julia we need to **install the libraries** as follows:

```
using Pkg
Pkg.add("LibraryName")
```

To **load the libraries**, we use the command `using`.

Here we will be loading a bunch of libraries so that we can load and use the data (`DataFrames`, `CSV`) and make some nice plots (`Plots`, `Binscatters`). We will also be running some fixed-effects regressions (`FixedEffectModels`).

```
begin
    using DataFrames
    using CSV
    using Plots
    using Binscatters
    using FixedEffectModels
end
```

We load the data using the CSV syntax (`CSV.read`) into a data frame called `df`. `first(df,5)` gives us a snapshot of the data.

	year	month	day	hour	dayofweek	demand	demand_forecast	wind	wind_fc
1	2009	5	1	1	5	25.337	25.021	5.7625	5.427
2	2009	5	1	2	5	23.478	23.044	5.7461	5.441
3	2009	5	1	3	5	21.859	21.685	5.786	5.487
4	2009	5	1	4	5	20.931	20.408	5.9837	5.532
5	2009	5	1	5	5	20.371	19.586	6.0831	5.551

```
begin
    df = CSV.read("data_spain.csv", DataFrame)
    first(df, 5)
end
```

The impacts of wind: visual exploration

We will be plotting the **impacts of wind** on several outcomes of interest:

- Emissions
- Wholesale prices
- System costs
- Wholesale prices + system costs

We will be using the library `Binscatters` for plotting.




```
begin
    binscatter(df, @formula(emis_tCO2 ~ wind_forecast + demand + fe(year)), 10,
        seriestype = :scatterpath,
        title = "Wind reduces emissions",
        xlabel = "Wind forecast (GWh)",
        ylabel = "Hourly emissions (tons CO2)", label="Forecast");
    binscatter!(df, @formula(emis_tCO2 ~ wind + demand + fe(year)), 10,seriestype =
:scatterpath)
end
```




```
binscatter(df, @formula(wholesale_price ~ wind_forecast), 10,
        seriestype = :scatterpath,
        title = "Wind reduces wholesale prices (cannibalization effect)",
        xlabel = "Wind forecast (GWh)",
        ylabel = "Wholesale price (EUR/MWh)")
```




```
binscatter(df, @formula(system_costs ~ wind_forecast + demand + fe(year)), 10,
        seriestype = :scatterpath,
        title = "Wind increases system costs (intermittency)",
        xlabel = "Wind forecast (GWh)",
        ylabel = "System costs (EUR/MWh)")
```




```
begin
    df.total_price = df.wholesale_price + df.system_costs
    binscatter(df, @formula(total_price ~ wind_forecast), 10,
        seriestype = :scatterpath,
        title = "But still reduces overall costs",
        xlabel = "Wind forecast (GWh)",
        ylabel = "Wholesale price + system costs (EUR/MWh)")
end
```

Wind endogeneity

One can estimate the effects of wind using a regression framework. However, it is important to keep in mind that wind production can be endogenous.

In moments of very high forecasted wind, it is often the case that wind is discarded. This can create an endogeneity problem.




```
scatter(df.wind_forecast, df.wind, xlabel="Forecasted wind", ylabel="Wind",
legend=false, title="Discarded wind")
```

We can examine the endogeneity problem in the context of assessing the impact of wind on reliability and other congestion costs ("system costs").

In days of very high wind, measured wind production could be lower than expected, leading to downward bias in our estimates: a difficult day with lots of wind appears as a day with low levels of wind in the data.

To address this issue, one can instrument wind production with forecasted wind.

We will be running these regressions using the `FixedEffectModels` library.

```
                          Fixed Effect Model
=======================================================================
Number of obs:                  78731   Degrees of freedom:           23
R2:                             0.228   R2 Adjusted:               0.228
F-Stat:                       2762.59   p-value:                   0.000
R2 within:                      0.034   Iterations:                    6
=======================================================================
system_costs | Estimate  Std.Error t value Pr(>|t|) Lower 95% Upper 95%
-----------------------------------------------------------------------
wind         | 0.187218 0.00356196 52.5603    0.000  0.180236  0.194199
=======================================================================
```

```
reg(df, @formula(system_costs ~ wind + fe(year) + fe(month)))
```

```
                         IV Fixed Effect Model
=======================================================================
Number of obs:                  78730   Degrees of freedom:           23
R2:                             0.227   R2 Adjusted:               0.227
F-Stat:                       3524.17   p-value:                   0.000
F-Stat (First Stage):       1.15385e6   p-value (First Stage):     0.000
R2 within:                      0.033   Iterations:                    6
=======================================================================
system_costs | Estimate  Std.Error t value Pr(>|t|) Lower 95% Upper 95%
-----------------------------------------------------------------------
wind         |  0.21866 0.00368332 59.3648    0.000   0.21144  0.225879
=======================================================================
```

```
reg(df, @formula(system_costs ~ (wind ~ wind_forecast) + fe(year) + fe(month)))
```

A policy change

The paper explores the **role of market design** in affecting the value of wind. The market moved away from subsidies that are paid based on production to subsidies that are based on installed capacity (subject to minimum performance requirments).

In the wholesale market, this implies that renewables no longer have an incentive to produce when prices are very low, e.g., as in California or Texas, in which prices are often zero or negative.

We will split the data in two to examine the change in the distribution of wholesale prices around the policy change.

Note: This is an event study, so there are other changes happening in the market. The idea here is to show major effects of the policy, but a proper quantification requires more explicit control of confounders.




```
begin
    df.policy = (((df.year .> 2014) .| ((df.year.==2014) .& (df.month .> 5))));
    dfpre = subset(df, :policy => ByRow(==(0)));
    dfpost = subset(df, :policy => ByRow(==(1)));

    # focus only on 2013-2015 data
    dfpre = subset(dfpre, :year => ByRow(>(2012)));
    dfpost = subset(dfpost, :year => ByRow(<(2016)));

    histogram(dfpre.wholesale_price, label="Pre")
    histogram!(dfpost.wholesale_price, label="Post")
end
```

The policy change appeared to reduce system costs in the market. This could be due to the challenges of dispatching the market in the presence of zero prices, which lead to several strategic distortions.

We can plot system costs before and after the change.




```
begin
    binscatter(dfpre, @formula(system_costs ~ wind_forecast), 10,
        label="Pre", seriestype = :linearfit)
    binscatter!(dfpost, @formula(system_costs ~ wind_forecast), 10,
        label="Post", seriestype = :linearfit)
end
```

Consumers were still worse off due to the increase in prices. Wind price reduction effect diminished.




```
begin
    binscatter(dfpre, @formula(total_price ~ wind_forecast), 10,
        label="Pre", seriestype = :linearfit)
    binscatter!(dfpost, @formula(total_price ~ wind_forecast), 10,
        label="Post", seriestype = :linearfit)
end
```

Follow-up exercises

1. What is the environmental benefit of wind power in this market per unit of wind? Try to quantify that by regressing emissions on wind and converting it to a monetary amount using a valuation for emissions reductions.
2. What is the correlation of wind and demand? How could that affect the valuation of wind power?